
Maple-assisted derivation of recurrence for A183334

Robert Israel

18 May 2020

There are  possible configurations for two adjacent rows, but only those where each 1 is 
adjacent to at most two 0's is allowed.  The following code computes the allowed configurations, of 
which there turn out to be 2584.  

Configs:= NULL: nConfigs:= 0:
for n from 0 to 4095 do
  v:= convert(4096+n,base,2)[1..12];
  good:= true:
  for i from 1 to 12 do
    if v[i] = 1 then
      t:= 0: 
      if i mod 6 <> 1 then t:= t + (1-v[i-1]) fi;
      if i mod 6 <> 0 then t:= t + (1-v[i+1]) fi;
      if i <= 6 then t:= t + 1-v[i+6] else t:= t + 1-v[i-6] fi;
      if t > 2 then good:= false; break fi
    fi
  od;
  if good then Configs:= Configs, v; nConfigs:= nConfigs+1; fi
od:
Configs:= [Configs]:
nConfigs;

2584

Let  be the  transition matrix such that  if the bottom two rows of a array 
could be in configuration  while the top two rows are in configuration  (thus the middle row is both 
the top row of configuration  and the bottom row of configuration , and every 1 in the middle row has
exactly two neighbours that are 0's).  The following code computes it.
q:= proc(i,j) local Ci, Cj,k;
   Ci:= Configs[i]; Cj:= Configs[j];
   if Ci[1..6] <> Cj[7..12] then return 0 fi;
   if Ci[1] = 1 and Ci[7]+Ci[2]+Cj[1] <> 1 then return 0 fi;
   if Ci[6] = 1 and Ci[5]+Ci[12]+Cj[6] <> 1 then return 0 fi;
   for k from 2 to 5 do 
     if Ci[k] = 1 and Ci[k-1]+Ci[k+1]+Ci[k+6]+Cj[k] <> 2 then 
return 0 fi
   od;
   1
end proc: 
T:= Matrix(nConfigs,nConfigs, q):

Let  be the column vector with 1's corresponding to configurations that could be the first two rows of 
an array, in that each 1 in the first row is adjacent to exactly two 0's, 0 otherwise, and similarly v the 
column vector with 1's corresponding to configurations that could be the bottom two rows.
Then  for .

u:= Vector(nConfigs, proc(i) local Ci,k;
  Ci:= Configs[i];
  if Ci[1] = 1 and Ci[2]+Ci[7] <> 0 then return 0 fi;

  if Ci[6] = 1 and Ci[5]+Ci[12] <> 0 then return 0 fi:
  for k from 2 to 5 do if Ci[k] = 1 and Ci[k-1]+Ci[k+1]+Ci[k+6] 
<> 1 then return 0 fi od:
  1 end proc):
v:=  Vector(nConfigs, proc(i) local Ci,k;
  Ci:= Configs[i];
  if Ci[7] = 1 and Ci[1]+Ci[8] <> 0 then return 0 fi;
  if Ci[12] = 1 and Ci[11]+Ci[6] <> 0 then return 0 fi:
  for k from 8 to 11 do if Ci[k] = 1 and Ci[k-1]+Ci[k+1]+Ci[k-6] 
<> 1 then return 0 fi od:
  1 end proc):

To check, here are the first few entries of our sequence.  For future use, I compute  for some 
values of 

U[0]:= u:
for n from 1 to 100 do U[n]:= T . U[n-1] od:
seq(v^%T . U[n], n = 0 .. 18);

To find a recurrence, we check when the vectors  fail to be linearly independent.
M:= U[0]:
for r from 1 do
  M:= <M|U[r]>;
  if LinearAlgebra:-Rank(M) < r+1 then printf("dependent for r=
%d\n",r); break fi
od:

dependent for r=98
The recurrence corresponds to a vector in the null space of the matrix formed by those columns.

R:= op(1,LinearAlgebra:-NullSpace(M)):

convert(R,list);

If this vector is , it means that  so that  

for  Here is the recurrence:
n:= 'n':

a(n) = sort(- add(R[i+1]/R[r+1] * a(n+i-r),i=0..r-1),[seq(a(n-i),
i=1..r)]);

We have proven it for , but it turns out to also be true for .


